********************************************************************************
*
* Exporting inequality: US investors and the Americanization of executive pay 
* in the United Kingdom
* 
* Lukas Linsi, Jonathan Hopkin & Pascal Jaupart
*
* Review of International Political Economy, DOI: 10.1080/09692290.2021.2004440
*
********************************************************************************

/*
This do-file reproduces the regression tables shown in the RIPE 2021 journal 
article publication based on *FIRM-level* data.

NB: The remumeration data being confidential, access to the analysis dataset
is not provided. 

Do-file outline:
	1. Data preparation 
	2. Main text regression tables
	3. Appendix tables
	
*/


*****************************************
********** 1. Data preparation **********
*****************************************

***** Loading data 
	clear
	set more off 
		global root " " // add own path to dataset
		use "$root\company_panel.dta"
	
	
***** Sort and Panel data	
	sort BoardID year
		xtset BoardID year


***** Ownership variables 
	gen share_foreign = US_shareholdertotal + NON_US_shareholdertotal
		replace share_foreign = 100 if share_foreign>100
		
	gen usshare_profit = US_shareholdertotal * stockprice
	gen usshare_union = US_shareholdertotal *  uniondensity 	
	
	
***** Economic Sectors
	
	// Financial services sector
	gen service_finance=0
		replace service_finance=1 if Sector=="Banks"
		replace service_finance=1 if Sector=="Insurance"
		replace service_finance=1 if Sector=="Investment Companies"
		replace service_finance=1 if Sector=="Life Assurance"	
		replace service_finance=1 if Sector=="Private Equity"
		replace service_finance=1 if Sector=="Real Estate"
		replace service_finance=1 if Sector=="Speciality & Other Finance"

	// Non-financial services
	gen service_nonfinance=0
		replace service_nonfinance=1 if Sector=="Business Services"
		replace service_nonfinance=1 if Sector=="Consumer Services"
		replace service_nonfinance=1 if Sector=="Food & Drug Retailers"
		replace service_nonfinance=1 if Sector=="General Retailers"
		replace service_nonfinance=1 if Sector=="Health"
		replace service_nonfinance=1 if Sector=="Leisure & Hotels"
		replace service_nonfinance=1 if Sector=="Media & Entertainment"
		replace service_nonfinance=1 if Sector=="Publishing"
		replace service_nonfinance=1 if Sector=="Software & Computer Services"
		replace service_nonfinance=1 if Sector=="Telecommunication Services"
		replace service_nonfinance=1 if Sector=="Transport"
		replace service_nonfinance=1 if Sector=="Utilities - Other"
	
	// Primary resources
	gen primaryresources_related=0
		replace primaryresources_related=1 if Sector=="Forestry & Paper"
		replace primaryresources_related=1 if Sector=="Mining"
		replace primaryresources_related=1 if Sector=="Oil & Gas"
		replace primaryresources_related=1 if Sector=="Renewable Energy"
		replace primaryresources_related=1 if Sector=="Tobacco"
	
	// General industry
	gen industry=0
		replace industry=1 if Sector=="Beverages"	
		replace industry=1 if Sector=="Chemicals"	
		replace industry=1 if Sector=="Clothing  Leisure and Personal Products"	
		replace industry=1 if Sector=="Construction & Building Materials"	
		replace industry=1 if Sector=="Containers & Packaging"	
		replace industry=1 if Sector=="Diversified Industrials"	
		replace industry=1 if Sector=="Electricity"	
		replace industry=1 if Sector=="Food Producers & Processors"	
		replace industry=1 if Sector=="Household Products"	

	// High-Tech sector	
	gen high_tech=0
		replace high_tech=1 if Sector=="Aerospace & Defence"
		replace high_tech=1 if Sector=="Automobiles & Parts"
		replace high_tech=1 if Sector=="Information Technology Hardware"
		replace high_tech=1 if Sector=="Pharmaceuticals and Biotechnology"
		replace high_tech=1 if Sector=="Electronic & Electrical Equipment"	
		replace high_tech=1 if Sector=="Engineering & Machinery"	
	
	
***** Variable labels
	
	// firm controls 
	label var logoperaterev "Operating revenue (log)"
	label var profitmargin "Profit margin (%)"
	label var solvency "Solvency ratio"
	label var stockprice "Stock price change"
	label var bvdindepcat "BvD independence"
	label var uniondensity "Union density"
	
	// pay outcome variables
	label var median_rem_exed "Median remuneration executive directors (in log)"		
	label var mean_rem_exed "Mean remuneration executive directors (in log)"		

	// foreign ownership
	label var share_foreign "Foreign ownership (%)"
	label var US_shareholdertotal "US ownership (%)"
	label var NON_US_shareholdertotal "Non-US ownership (%)"	

	// interaction terms
	label var usshare_profit "US ownership (%) x Profit margin (%)"
	label var usshare_union "US ownership (%) x Union density"

	

****************************************************	
********** 2. Main text regression tables **********
****************************************************
	
***** Table 2: The impact of foreign ownership on executive pay
	
	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
	local sector_trends "sectorid2_trend-sectorid39_trend"
	
	// column 1
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)
	
	// column 2
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year, fe cluster(BoardID)
		
	// column 3	
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
		
		
***** Table 3: US ownership determinants

	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"

	// column 1
	xtreg US_shareholdertotal logoperaterev i.year, fe cluster(BoardID)

	// column 2	
	xtreg US_shareholdertotal profitmargin i.year, fe cluster(BoardID)

	// column 3	
	xtreg US_shareholdertotal solvency i.year, fe cluster(BoardID)

	// column 4	
	xtreg US_shareholdertotal stockprice i.year, fe cluster(BoardID)

	// column 5	
	xtreg US_shareholdertotal bvdindepcat i.year, fe cluster(BoardID)

	// column 6	
	xtreg US_shareholdertotal uniondensity i.year, fe cluster(BoardID)

	// column 7	
	xtreg US_shareholdertotal `covars' i.year, fe cluster(BoardID)
				
		
***** Table 5: Relationship in different industry branches

	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"

	// Financial services - columns 1 and 2
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if service_finance==1, fe cluster(BoardID)
		
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if service_finance==1, fe cluster(BoardID)
					
	// Non-financial services - columns 3 and 4
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if service_nonfinance==1, fe cluster(BoardID)
		
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if service_nonfinance==1, fe cluster(BoardID)

	// Primary resources - columns 5 and 6
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if primaryresources_related==1, fe cluster(BoardID)
		
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if primaryresources_related==1, fe cluster(BoardID)

	// General industry - columns 7 and 8
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if industry==1, fe cluster(BoardID)
		
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if industry==1, fe cluster(BoardID)

	// High tech - columns 9 and 10
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if high_tech==1, fe cluster(BoardID)
		
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if high_tech==1, fe cluster(BoardID)
	

***** Table 7: American ownership and company performance

	preserve

		keep if median_rem_exed!=.
		local covars "bvdindepcat uniondensity"
		local sector_trends "sectorid2_trend-sectorid39_trend"
	 
		// revenue - columns 1 and 2
		xtreg logoperaterev US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

		xtreg logoperaterev US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
				
		// profit margin - columns 3 and 4
		xtreg profitmargin US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

		xtreg profitmargin US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
				
		// solvency - columns 5 and 6
		xtreg solvency US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

		xtreg solvency US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)

		// stockprice - columns 7 and 8
		xtreg stockprice US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

		xtreg stockprice US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)

	restore	
	
	
***** Table 8: Probability of hiring American citizens

	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
	local sector_trends "sectorid2_trend-sectorid39_trend"

	// column 1
	xtreg uscitizen US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

	// column 2
	xtreg uscitizen US_shareholdertotal NON_US_shareholdertotal `covars' i.year, fe cluster(BoardID)

	// column 3
	xtreg uscitizen US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
		
	
	
****************************************
********** 3. Appendix tables **********
****************************************		

***** Table A6: Summary statistics
	sum median_rem_exed mean_rem_exed US_shareholdertotal NON_US_shareholdertotal ///
			logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity uscitizen ///
			service_finance service_nonfinance primaryresources_related industry high_tech	

	estpost summarize median_rem_exed mean_rem_exed US_shareholdertotal NON_US_shareholdertotal ///
					 logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity uscitizen ///
					 service_finance service_nonfinance primaryresources_related industry high_tech 

	esttab, cells("count mean sd min max")

	
***** Table A8: Adding interaction terms

	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
	local sector_trends "sectorid2_trend-sectorid39_trend"

	// Column 1 
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)

	// Column 2
	xtreg median_rem_exed US_shareholdertotal usshare_profit NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
			
	// Column 3		
	xtreg median_rem_exed US_shareholdertotal usshare_union NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
		
		
***** Table A9: Balanced panel
			
	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
	local sector_trends "sectorid2_trend-sectorid39_trend"

	// Column 1 	
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year if countobs==8, fe cluster(BoardID)

	// Column 2	
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year `sector_trends' if countobs==8, fe cluster(BoardID)

	// Column 3	
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year if countobs==8, fe cluster(BoardID)

	// Column 4	
	xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends' if countobs==8, fe cluster(BoardID)


***** Table A10: Mean remuneration 
	
	local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
	local sector_trends "sectorid2_trend-sectorid39_trend"

	// Column 1 	
	xtreg mean_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)

	// Column 2		
	xtreg mean_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year `sector_trends', fe cluster(BoardID)

	// Column 3		
	xtreg mean_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year, fe cluster(BoardID)

	// Column 4		
	xtreg mean_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
		
		
***** Table A13: Excluding firms hiring US citizen EDs		

	preserve

		egen us_firm = max(uscitizen), by(BoardID)
			ta us_firm
			keep if us_firm==0		
				
		local covars "logoperaterev profitmargin solvency stockprice bvdindepcat uniondensity"
		local sector_trends "sectorid2_trend-sectorid39_trend"

		// Column 1		
		xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year, fe cluster(BoardID)		
		
		// Column 2
		xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal i.year `sector_trends', fe cluster(BoardID)

		// Column 3			
		xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year, fe cluster(BoardID)

		// Column 4		
		xtreg median_rem_exed US_shareholdertotal NON_US_shareholdertotal `covars' i.year `sector_trends', fe cluster(BoardID)
			
	restore			
		

// End.

	